

    

        ProcessTree

        v0.3.0



    



  

    Table of contents

    
      


          	Examples
            


            	Environment Variable Example



            

          




  	Modules
    

    	ProcessTree


    

  



      

    


  

    
Environment Variable Example
    

This example demonstrates how ProcessTree can be used in tests to provide custom values for 
environment variables, while preserving async: true.
The example is a Phoenix app with a single LiveView that serves the site's index page. The 
LiveView uses an environment variable called cutoff_date to customize the appearance of the 
index page. When a viewer hits the page before the cutoff date, they see this message:
    Welcome! You've made the cutoff date :-)

When a viewer hits the page after the cutoff date, they see this message:
    Sorry! You've missed the cutoff date :-(

For "real-life" usage when MIX_ENV=dev or MIX_ENV=prod, the cutoff_date is set in runtime.exs as a hardcoded 
Application environment variable:
config :environment_variable_example, cutoff_date: ~D[2024-01-01]
In our tests of the LiveView, we override the value on a test-specific basis by writing the desired
custom value to the process dictionary of the ExUnit test pid:
test "when the cutoff date has not passed, shows a 'welcome' message", %{conn: conn} do
  tomorrow = Date.add(Date.utc_today(), 1)

  # set the cutoff date to tomorrow by writing the value
  # to the process dictionary of the ExUnit test pid
  Process.put(:cutoff_date, tomorrow)

  {:ok, _show_live, html} = live(conn, ~p"/welcome")

  assert html =~ "Welcome!"
end

test "when cutoff date has passed, shows a 'sorry' message", %{conn: conn} do
  yesterday = Date.add(Date.utc_today(), -1)

  # set the cutoff date to yesterday by writing the value
  # to the process dictionary of the ExUnit test pid
  Process.put(:cutoff_date, yesterday)

  {:ok, _show_live, html} = live(conn, ~p"/welcome")

  assert html =~ "Sorry!"
end
Since this custom value is scoped to individual ExUnit tests, rather than to the global Application environment, the tests are safe for async: true.
The code for the LiveView is below. In the cutoff_date() function, the LiveView uses ProcessTree to look for a customized value. If no custom value is found (as when running in dev or prod), then the LiveView uses the cutoff_date retrieved from the Application environment, as configured in runtime.exs 
defmodule EnvironmentVariableExampleWeb.WelcomeLive.Show do
  @moduledoc false
  use EnvironmentVariableExampleWeb, :live_view

  @impl true
  def render(assigns) do
    message = case Date.before?(Date.utc_today(), cutoff_date()) do
      true ->
        "Welcome! You've made the cutoff date :-)"

      false ->
        "Sorry! You've missed the cutoff date :-("
    end

    assigns = assign(assigns, :message, message)

    ~H"""
    <h1 style='padding: 3em'><%= @message %></h1>
    """
  end

  defp cutoff_date() do
    # read the default cutoff_date from the Application environment
    app_env_cutoff_date = Application.get_env(:environment_variable_example, :cutoff_date)

    # ProcessTree will look for a customized cutoff_date value in the process dictionaries
    # of this process and its ancestors.
    #
    # When running our tests, ProcessTree will find and return the customized value that
    # we have inserted into the process dictionary of the ExUnit test pid, which is an
    # ancestor of this LiveView process.
    #
    # When running in production, ProcessTree will not find a value in any ancestor
    # dictionaries, and it will return the default value after caching it in the process
    # dictionary of the current process.
    ProcessTree.get(:cutoff_date, default: app_env_cutoff_date)
  end
end
During tests, the LiveView "sees" the custom value because ProcessTree looks up the process ancestry hierarchy to 
find the value. The LiveView process is spawned by the ExUnit test pid (indirectly, via the ExUnit test supervisor), meaning that the test pid is an ancestor of the LiveView. ProcessTree eventually finds the custom value in the process dictionary of the test pid.
In production, ProcessTree returns the default value obtained from the Application environment.



  

    
ProcessTree 
    



      
ProcessTree is a module for avoiding global state in Elixir applications. 

  
    
  
  Motivation


It's common in Elixir apps to rely on global references or global variables for locating application services or configuration parameters. This presents problems when we test our code. We need to work around our global data, for example by setting/resetting environment variables and using async: false.
ProcessTree solves the problem by localizing data to a particular branch of the Elixir process tree. When testing with ExUnit, each ExUnit test process, and all processes spawned by the test process, can use get/2 to see their own private copy of the data of interest.

  
    
  
  How to use get()


get/2 can be used to make child processes "see" values stored in the process dictionaries of their ancestor processes. This is useful, for example, in ExUnit tests that spawn processes, such as tests that start LiveViews and GenServers.
To make data visible to child processes via get/2, we first put the data into the process dictionary of an ancestor process - in this case, an ExUnit test pid:
test "some test that starts a GenServer" do
  # ...
  # add the data of interest to the process dictionary of the test pid
  Process.put(:some_key, some_value)

  # The GenServer process started here can use ProcessTree.get() to see
  # the value we've bound to :some_key
  server = MyGenserver.start_link()
  # ...
end

  
    
  
  Example use case


Customizing environment variables in ExUnit tests while preserving async: true.

  
    
  
  Smoothing over process ancestry complications


OTP 25 introduced the ability to find the parent of a process via Process.info/2. Prior to OTP 25, it was possible to find the parent of a process only for specific processes such as Task, GenServer, Agent, and Superivsor.
Even under OTP 25+, Process.info/2 is only useful for processes that are still alive, meaning it can't be used, for example, to find the grandparent of a process if the parent of the process has died. 
ProcessTree accounts for all complicating factors. Each of the functions exposed by ProcessTree will return the most complete answer possible, regardless of how the processes in the ancestry hierarchy are started or managed, regardless of which OTP version is in use, and so on.
Put another way, ProcessTree is a "no-judgments zone". If you're following recommended guidelines for starting/running Elixir processes, ProcessTree will almost certainly meet your needs. But even in situations typically considered inadvisable or totally crazy, ProcessTree will do its very best to provide meaningful answers.

      


      
        Summary


  
    Functions
  


    
      
        get(key, opts \\ [])

      


        Starting with the calling process, recursively looks for a value for key in the
process dictionaries of the calling process and its known ancestors/callers.



    


    
      
        get_from(pid, key)

      


        Recursively looks for a value for key in the process dictionaries of pid and its known
ancestors/callers.



    


    
      
        known_ancestors(pid)

      


        Returns a list of the known ancestors of pid, from "newest" to "oldest".



    


    
      
        parent(pid)

      


        Returns the parent of pid, if the parent is known.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    get(key, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec get(
  term(),
  keyword()
) :: term()


      


Starting with the calling process, recursively looks for a value for key in the
process dictionaries of the calling process and its known ancestors/callers.
If a particular ancestor process has died, it looks up to the next ancestor, if possible.
Returns the first non-nil value found in the tree. If no value is found, returns nil.
By default, if a non-nil value is found in the tree, the value is cached in the dictionaries
of the calling process and any intermediate processes traversed in the search. Thereafter,
subsequent calls to get() with the same key will return the cached value. This behavior
can be disabled by passing cache: false as an option.

  
    
  
  Parents, $ancestors and $callers


Erlang provides two mechanisms
for finding the ancestors of a process: :erlang.process_info(pid, :parent), available
since OTP 25, and the $ancestors process dictionary key, which only applies to processes
managed by OTP.
In addition to these, Elixir uses the $callers
dictionary key to track the chain of processes that initiate a supervised task.
get/2 considers all three mechanisms when navigating the ancestry of a process. Priority is first
given to the parent/$ancestors hierarchy. Given a process and a key, if neither the process itself
nor any parents/$ancestors have a dictionary value for the key, get/2 asks the process if it has
any $callers. If it does, get/2 will look up to the first caller, look in its dictionary, then
if necessary examine the caller's parents/$ancestors and $callers, with priority again given to
the parent/$ancestors hierarchy.

  
    
  
  Remote ancestors


In cases where it is called within a process started by a remote node, for example via the
Node.spawn functions, get() will examine only the processes local to its node. Its
search stops when it encounters a remote pid.

  
    
  
  Options


	:cache - If false, the value found in the tree will not be cached in the dictionaries
of the calling process or any intermediate processes traversed in the search. Defaults to true.

	:default - When no value is found in the tree, get() will return the default value
if one is provided. If caching is enabled, the default value will be cached in the
dictionary of the calling process.
Default values might typically be read from the Application environment. For example:
default_value = Application.get_env(:my_app, :some_key)
value = ProcessTree.get(:some_key, default: default_value)

	:lazy_default - A zero-arity function that will be called to obtain a default value
when no value is found in the tree. This is useful when it is impossible or undesirable
to compute the default value unless/until it is actually requested.
If caching is enabled, the default value will be cached in the dictionary of the
calling process.
:default and :lazy_default are mutually exclusive; get/2 will raise an ArgumentError
if both options are provided.
Examples:
value = ProcessTree.get(:some_key, lazy_default: fn -> compute_me_later() end)
value = ProcessTree.get(:some_key, lazy_default: &compute_me_later/0)



  



  
    
      
      Link to this function
    
    get_from(pid, key)


      
       
       View Source
     


  


  

      

          @spec get_from(pid(), term()) :: term()


      


Recursively looks for a value for key in the process dictionaries of pid and its known
ancestors/callers.
If a particular ancestor process has died, it looks up to the next ancestor, if possible.
Returns the first non-nil value found in the tree. If no value is found, returns nil.
In terms of navigating the tree, get_from/2 behaves identically to get/2.
Unlike get/2, get_from/2 is not expected to be used in production code, but rather
only in less-common testing situations such as the one described in this
github issue.
Accordingly, get_from/2 does not have options for caching values or returning default values, as those options
lose their benefit outside of production code.

  



  
    
      
      Link to this function
    
    known_ancestors(pid)


      
       
       View Source
     


  


  

      

          @spec known_ancestors(pid()) :: [pid() | atom()]


      


Returns a list of the known ancestors of pid, from "newest" to "oldest".
The set of ancestors that is "known" depends on factors including:
	The OTP major version the code is running under. (OTP 25 introduced
new functionality for tracking ancestors.)
	Whether the process and its ancestors are running in a supervision tree
	Whether ancestor processes are still alive
	Whether the given process and its ancestors were started via raw spawn or
were instead started as Tasks, Agents, GenServers or Supervisors

ProcessTree takes these factors and more into account and produces the most complete
list of ancestors that is possible.
In the mainline case - running under a supervision tree, as recommended - known_ancestors/1 will
return a list that contains, at minimum, all of the ancestor Supervisors in the tree
as well as the ancestor of the initial/topmost Supervisor.
When running under OTP 25 and later, the list will also include all additional ancestors,
up to and including the :init process (PID<0.0.0>), provided that the additional ancestor
processes are still alive.
List items will be pids in all but the most unusual of circumstances. For example, if a GenServer
is spawned by parent/grandparent GenServers that have registered names, and the parent GenServer dies,
then the parent & grandparent may be represented in the list of the child's known ancestors using
the their registered names - atoms, rather than pids. Precise behavior depends on OTP major version.
In cases where it is called within a process started by a remote node, for example via the
Node.spawn functions, the list returned by known_ancestors() will contain only the pids
belonging to the local node. known_ancestors() ends its search when it encounters a remote
pid in the ancestry hierarchy.

  



  
    
      
      Link to this function
    
    parent(pid)


      
       
       View Source
     


  


  

      

          @spec parent(pid()) :: pid() | atom()


      


Returns the parent of pid, if the parent is known.
Returns :unknown if the parent is unknown, or if the parent belongs to a remote node.
Returns :undefined if pid represents the :init process (PID<0.0.0>).
If pid is part of a supervision tree, the parent will be known regardless of any other factors.
If the parent is known, the return value will be a pid in all but the most unusual of
circumstances. See known_ancestors/1 for dicussion.

